
  CMS819V0 MP   2012







  MEASURE DETAILS

HOSPITAL HARM - OPIOID-RELATED ADVERSE EVENTS

Episode of Care Based Measure

This measure assesses the proportion of inpatient hospital encounters where patients ages 18 years of age or
older have been administered an opioid medication, subsequently suffer the harm of an opioid-related adverse
event, and are administered an opioid antagonist (naloxone) within 12 hours. This measure excludes opioid
antagonist (naloxone) administration occurring in the operating room setting.

SHOW ALL RESULTS

Initial Population:

define "Initial Population": 
  "Encounters With Opioid Administration"

SHOW RESULT

Denominator:

define "Denominator": 
  "Initial Population"

SHOW RESULT

Numerator:

define "Numerator": 
  "Encounters With Opioid Antagonist Administration Outside of Operating Room and 
Within 12 Hrs After Opioid"

 ()

SHOW RESULT

Definitions:

define "Opioid Administration": 
  ["Medication, Administered": "Opioids, All"]

SHOW RESULT

define "Qualifying Encounters": 
  ["Encounter, Performed": "Encounter Inpatient"] InpatientEncounter 
    with ["Patient Characteristic Birthdate": "Birth date"] BirthDate 
      such that InpatientEncounter.relevantPeriod ends during "Measurement Period" 
        and Global."CalendarAgeInYearsAt" ( BirthDate.birthDatetime, start of 
Global."HospitalizationWithObservation" ( InpatientEncounter ) ) >= 18

SHOW RESULT

define "Encounters With Opioid Administration": 
  "Qualifying Encounters" QualifyingEncounter 
    with "Opioid Administration" Opioid 
      such that Opioid.relevantPeriod during Global."HospitalizationWithObservation" ( 
QualifyingEncounter )

SHOW RESULT

define "Encounters With Opioid Antagonist Administration Outside of Operating Room and 
Within 12 Hrs After Opioid": 
  from 
    ["Medication, Administered": "Opioid Antagonist"] OpioidAntagonistAdministration, 
    "Opioid Administration" OpioidAdministration, 
    "Denominator" EncounterWithQualifyingAge  // Opioid antagonist not administered in 
operating room  
    where not exists ( EncounterWithQualifyingAge.facilityLocations EncounterLocation 
        where EncounterLocation.code = "Operating Room/Suite" 
          and OpioidAntagonistAdministration.relevantDatetime during 
EncounterLocation.locationPeriod 
    )  // AND, opioid antagonist administered after start of encounter, and opioids 
administered before opioid antagonist 
      and ( OpioidAntagonistAdministration.relevantDatetime during 
Global."HospitalizationWithObservation" ( EncounterWithQualifyingAge ) 
          and OpioidAdministration.relevantPeriod during 
Global."HospitalizationWithObservation" ( EncounterWithQualifyingAge ) 
          and OpioidAdministration.relevantPeriod ends 12 hours or less before 
OpioidAntagonistAdministration.relevantDatetime 
      ) 
    return EncounterWithQualifyingAge

SHOW RESULT

Functions:

define function "LengthInDays"(Value Interval<DateTime> ): 
  difference in days between start of Value and end of Value

define function "CalendarAgeInDaysAt"(BirthDateTime DateTime, AsOf DateTime ): 
  days between ToDate(BirthDateTime)and ToDate(AsOf)

define function "CalendarAgeInDays"(BirthDateTime DateTime ): 
  CalendarAgeInDaysAt(BirthDateTime, Today())

define function "CalendarAgeInMonthsAt"(BirthDateTime DateTime, AsOf DateTime ): 
  months between ToDate(BirthDateTime)and ToDate(AsOf)

define function "CalendarAgeInMonths"(BirthDateTime DateTime ): 
  CalendarAgeInMonthsAt(BirthDateTime, Today())

define function "CalendarAgeInYearsAt"(BirthDateTime DateTime, AsOf DateTime ): 
  years between ToDate(BirthDateTime)and ToDate(AsOf)

define function "CalendarAgeInYears"(BirthDateTime DateTime ): 
  CalendarAgeInYearsAt(BirthDateTime, Today())

define function "HospitalizationLocations"(Encounter "Encounter, Performed" ): 
  Encounter Visit 
    let EDVisit: Last(["Encounter, Performed": "Emergency Department Visit"] LastED 
        where LastED.relevantPeriod ends 1 hour or less on or before start of 
Visit.relevantPeriod 
        sort by  
        end of relevantPeriod 
    ) 
    return if EDVisit is null then Visit.facilityLocations  
      else flatten { EDVisit.facilityLocations, Visit.facilityLocations }

define function "EmergencyDepartmentArrivalTime"(Encounter "Encounter, Performed" ): 
  start of First(("HospitalizationLocations"(Encounter))HospitalLocation 
      where HospitalLocation.code in "Emergency Department Visit" 
      sort by start of locationPeriod 
  ).locationPeriod

define function "Hospitalization"(Encounter "Encounter, Performed" ): 
  Encounter Visit 
    let EDVisit: Last(["Encounter, Performed": "Emergency Department Visit"] LastED 
        where LastED.relevantPeriod ends 1 hour or less on or before start of 
Visit.relevantPeriod 
        sort by  
        end of relevantPeriod 
    ) 
    return Interval[Coalesce(start of EDVisit.relevantPeriod, start of 
Visit.relevantPeriod), 
    end of Visit.relevantPeriod]

define function "HospitalAdmissionTime"(Encounter "Encounter, Performed" ): 
  start of "Hospitalization"(Encounter)

define function "HospitalArrivalTime"(Encounter "Encounter, Performed" ): 
  start of First(("HospitalizationLocations"(Encounter))HospitalLocation 
      sort by start of locationPeriod 
  ).locationPeriod

define function "HospitalizationLengthofStay"(Encounter "Encounter, Performed" ): 
  LengthInDays("Hospitalization"(Encounter))

define function "HospitalDepartureTime"(Encounter "Encounter, Performed" ): 
  end of Last(("HospitalizationLocations"(Encounter))HospitalLocation 
      sort by start of locationPeriod 
  ).locationPeriod

define function "HospitalDischargeTime"(Encounter "Encounter, Performed" ): 
  end of Encounter.relevantPeriod

define function "HospitalizationWithObservationAndOutpatientSurgeryService"(Encounter 
"Encounter, Performed" ): 
  Encounter Visit 
    let ObsVisit: Last(["Encounter, Performed": "Observation Services"] LastObs 
        where LastObs.relevantPeriod ends 1 hour or less on or before start of 
Visit.relevantPeriod 
        sort by  
        end of relevantPeriod 
    ), 
    VisitStart: Coalesce(start of ObsVisit.relevantPeriod, start of 
Visit.relevantPeriod), 
    EDVisit: Last(["Encounter, Performed": "Emergency Department Visit"] LastED 
        where LastED.relevantPeriod ends 1 hour or less on or before VisitStart
        sort by  
        end of relevantPeriod 
    ), 
    VisitStartWithED: Coalesce(start of EDVisit.relevantPeriod, VisitStart), 
    OutpatientSurgeryVisit: Last(["Encounter, Performed": "Outpatient Surgery Service"] 
LastSurgeryOP 
        where LastSurgeryOP.relevantPeriod ends 1 hour or less on or before 
VisitStartWithED 
        sort by  
        end of relevantPeriod 
    ) 
    return Interval[Coalesce(start of OutpatientSurgeryVisit.relevantPeriod, 
VisitStartWithED),  
    end of Visit.relevantPeriod]

define function "HospitalizationWithObservation"(Encounter "Encounter, Performed" ): 
  Encounter Visit 
    let ObsVisit: Last(["Encounter, Performed": "Observation Services"] LastObs 
        where LastObs.relevantPeriod ends 1 hour or less on or before start of 
Visit.relevantPeriod 
        sort by  
        end of relevantPeriod 
    ), 
    VisitStart: Coalesce(start of ObsVisit.relevantPeriod, start of 
Visit.relevantPeriod), 
    EDVisit: Last(["Encounter, Performed": "Emergency Department Visit"] LastED 
        where LastED.relevantPeriod ends 1 hour or less on or before VisitStart
        sort by  
        end of relevantPeriod 
    ) 
    return Interval[Coalesce(start of EDVisit.relevantPeriod, VisitStart),  
    end of Visit.relevantPeriod]

define function "ToDate"(Value DateTime ): 
  DateTime(year from Value, month from Value, day from Value, 0, 0, 0, 0, 
timezoneoffset from Value)

define function "HospitalizationWithObservationLengthofStay"(Encounter "Encounter, 
Performed" ): 
  "LengthInDays"("HospitalizationWithObservation"(Encounter))

define function "FirstInpatientIntensiveCareUnit"(Encounter "Encounter, Performed" ): 
  First((Encounter.facilityLocations)HospitalLocation 
      where HospitalLocation.code in "Intensive Care Unit" 
        and HospitalLocation.locationPeriod during Encounter.relevantPeriod 
      sort by start of locationPeriod 
  )

Unused Definitions:

TERMINOLOGY

Birth date
Direct Reference Code
N/A

Emergency Department Visit
2.16.840.1.113883.3.117.1.7.1.292
Draft

Encounter Inpatient
2.16.840.1.113883.3.666.5.307
Draft

Ethnicity
2.16.840.1.114222.4.11.837
Draft

Intensive Care Unit
2.16.840.1.113762.1.4.1029.206
Draft

ONC Administrative Sex
2.16.840.1.113762.1.4.1
Draft

Observation Services
2.16.840.1.113762.1.4.1111.143
Draft

Operating Room/Suite
Direct Reference Code
N/A

Opioid Antagonist
2.16.840.1.113762.1.4.1179.1
Draft

Opioids, All
2.16.840.1.113762.1.4.1196.226
Draft

Outpatient Surgery Service
2.16.840.1.113762.1.4.1110.38
Draft

Payer
2.16.840.1.114222.4.11.3591
Draft

Race
2.16.840.1.114222.4.11.836
Draft

 PASS

 PASS

 PASS

 PASS

 PASS

DENOMPass Encounter Ends on Start of MP

DENOMPass Opioid During Encounter

DENOMPass Turns 18 day of Encounter

IPPFail Age17 Start of Encounter

IPPFail Antagonist but no Opioid

  TEST PATIENTS  ()

100 PASS   18 /18

100

% COVERAGE

 PASS

 PASS

 PASS

 PASS

 PASS

 PASS

 PASS

 PASS

 PASS

 PASS

 PASS

 PASS

 PASS

IPPFail Encounter Ends After MP

IPPFail Inpatient Late

IPPass Age18EDObsEnc

NUMERFail Antagonist before Opioid Ends

NUMERFail Antagonist in Operating Room

NUMERFail Naloxone 24Hrs + 1 minute After Opioid

NUMERFail Naloxone after encounter ends

NUMERFail Opioid >12 hrs before Naloxone

NUMERPass Opioid <12 hrs before Naloxone

NUMERPass Opioid exactly 12 hrs before Naloxone

NUMERPass Two Encounters, Both Qualify

NUMERPass Two Encounters, Only 1 qualifies

NUMERPass Two Num Events, Count only one




























